сообщения
объединенного
института
ядерных
исследований
дубна

P11 - 12860

В.Гаджоков, Н.Богданова

ПРОГРАММЫ ПОСТРОЕНИЯ СИСТЕМЫ ПОЛИНОМОВ, ОРТОНОРМИРОВАННЫХ НА ПРОИЗВОЛЬНОМ ТОЧЕЧНОМ МНОЖЕСТВЕ

1979

P11 - 12860

В.Гаджоков, Н.Богданова

ПРОГРАММЫ ПОСТРОЕНИЯ СИСТЕМЫ ПОЛИНОМОВ, ОРТОНОРМИРОВАННЫХ НА ПРОИЗВОЛЬНОМ ТОЧЕЧНОМ МНОЖЕСТВЕ

Гаджоков В., Богданова Н. Р11 - 12860

Программы построения системы полиномов, ортонормированных на произвольном точечном множестве

Предложен алгоритм построения системы полиномов, ортонормированных на конечном дискретном множестве при произвольных положительных весах. Показана связь между мощностью порождающего множества и числом полиномов системы, а также ее полнота. Приведены три взаимосвязанные программы на ФОРТРАНЕ-IV, реализующие предложенный алгоритм. Даны численные результаты модельного примера, демонстрирующего точность работы алгоритма на малой ЭВМ ЕС-1010. Проведен дополнительный численный эксперимент по соблюдению условий ортонормированности при выполнении вычислений с мантиссой в 23 бит и при порядках генерированных полиномов от 0 до 40.

Работа выполнена в Лаборатории вычислительной техники и автоматизации ОИЯИ.

Сообщение Объединенного института ядерных исследований. Дубна 1979

Gadjokov V., Bogdanova N. P11 - 12860

Constructing of a Polynomial System Orthonormal over an Arbitrary Point Set

An algorithm is proposed which builds a system of polynimials orthonormal over a finite discrete set with arbitrary positive weights. The connection is shown between the potence of the generating set and the number of polynomials in the system; the completeness of the latter is also demonstrated. A package of three FORTRAN-IV subroutines is given which implements the algorithm proposed. The numerical results of a model example are reported in a simple case where these can be analytically checked. An additional numerical experiment is conducted and the orthonormalization is tested for 23-bit mantissas when the order of polynomials generated varies from 0 to 40.

The investigation has been performed at the Laboratory of Computing Techniques and Automation, JINR.

Communication of the Joint Institute for Nuclear Research. Dubna 1979

1. ВВЕДЕНИЕ

Системы ортогональных функций, удовлетворяющих условию

$$\int f_m(x)\, w(x)\, f_n(x)\, dx = h_m \delta_{mn} , \qquad /1.1/$$

известны давно, считаются достаточно хорошо изученными и широко используются как в аналитических выкладках, так и в практике численных расчетов. По некоторым типам таких систем существует обширная литература /см. монографии [1-7] /.

В настоящей работе свойства полиномов, ортонормированных на конечных точечных множествах, используются для построения системы таких полиномов при заданном множестве с произвольными положительными весами. В разделе 2 вводятся необходимые для этой цели определения и обсуждаются вкратце свойства этих полиномов. В разделах 3 и 4 соответственно изложен алгоритм построения системы ортонормированных полиномов при произвольных конечных положительных весах и приведены три взаимосвязанные программы на ФОРТРАНе-IV, реализующие предложенный алгоритм. Работа алгоритма проиллюстрирована на простом примере, где возможно и аналитическое выполнение расчетов, которые в более сложных случаях требуют использования ЭВМ.

2. СВОЙСТВА ПОЛИНОМОВ, ОРТОНОРМИРОВАННЫХ НА ДИСКРЕТНОМ МНОЖЕСТВЕ

Рассмотрим конечное действительное точечное множество

$$M: \{ x_i \}_{i=1,2,\dots,m} . \qquad /2.1/$$

каждому элементу x_i которого поставлен в соответствие конечный положительный вес w_i . Поскольку любое такое множество может быть: а/ упорядочено так, чтобы $x_i < x_{i+1}$, и б/ сведено к замкнутому отрезку $[-1, +1]$ при помощи линейного преобразования

$$x'_i = \frac{2x_i}{x_{max}-x_{min}} - \frac{x_{max}+x_{min}}{x_{max}-x_{min}}, \qquad /2.2/$$

то в дальнейшем, без потери общности, мы будем считать M упорядоченным на [-1, +1].

Обсудим свойства полиномов $P_j(x)$ на [-1,1] :

$$\left.\begin{aligned} P_j(x) &= \sum_{k=0}^{j} a_k^{(j)} x^k \\ a_j^{(j)} &\neq 0 \end{aligned}\right\}, \qquad /2.3/$$

которые по аналогии с /1.1/ удовлетворяют дискретному условию ортогональности

$$\sum_{i=1}^{m} P_j(x_i) w_i P_k(x_i) = h_j \delta_{jk} . \qquad /2.4/$$

Следуя [6, §3.15] и [8], будем называть систему полиномов типа /2.3/, удовлетворяющую /2.4/, ортогональной на дискретном множестве M . Если вдобавок потребовать, чтобы нормы h_j равнялись единице при любом возможном j,

$$h_j \equiv 1, \qquad /2.5/$$

то система эта становится ортонормированной.

Заметим, что, несмотря на дискретность условия ортогональности /2.4/, полиномы такой системы существуют и конечны при любом значении $x \in [-1,+1]$. Более того, если представить веса в виде функции

$$W(x) = \sum_{i=1}^{m} w_i \, \delta(x-x_i), \qquad /2.6/$$

то условия ортонормированности /2.4/ и /2.5/ можно записать как интеграл

$$\int P_j(x) W(x) P_k(x) \, dx = \delta_{jk}, \qquad /2.7/$$

что соответствует привычному виду /1.1/, причем интегрирование следует понимать в смысле Стилтьеса [6, с.91].

Потребуем, чтобы постулированные таким образом полиномы удовлетворяли общему для любых систем ортогональных полиномов рекуррентному соотношению [6, с.35], которое мы запишем в виде [8]

$$P_{i+1}(x) = c_{i+1} \, [(x-\alpha_{i+1}) P_i(x) - \beta_i P_{i-1}(x)], \qquad /2.8/$$

где c_{i+1} является нормирующим коэффициентом полинома $(i+1)$ -й степени, а все полиномы меньших степеней предполагаются нормированными; α_{i+1} и β_i - рекуррентные коэффициенты, не зависящие от аргумента x.

Взяв /2.8/ в точке x_k, помножив обе стороны равенства на $w_k P_j(x_k)$ и просуммировав по всем $k = 1, 2, \dots, m$ с учетом ортогональности /2.4/, получаем

$$\alpha_{i+1} = \sum_{k=1}^{m} x_k P_i^2(x_k) w_k \qquad /2.9/$$

и

$$\beta_i = \sum_{k=1}^{m} x_k P_{i-1}(x_k) w_k P_i(x_k). \qquad /2.10/$$

Таким же путем, учитывая как ортогональность /2.4/, так и нормировку /2.5/ при $j=i+1$, вычисляем нормирующий коэффициент

$$c_{i+1} = \Big[\sum_{k=1}^{m} x_k^2 P_i^2(x_k) w_k - \alpha_{i+1}^2 - (1-\delta_{oi}) \beta_i^2 \Big]^{-1/2}, \qquad /2.11/$$

а также устанавливаем связь

$$c_{i+1} = 1/\beta_{i+1} . \qquad /2.12/$$

Множитель $(1-\delta_{oi})$ включен в /2.11/ с тем, чтобы это соотношение имело место и при $i=0$ /см. ниже/.

Выражения /2.11/ и /2.12/ более удобны для расчета нормирующего коэффициента c_{i+1} и рекуррентного коэффициента β_{i+1}, т.к. требуют только значений полинома предыдущего порядка i, тогда как /2.10/ требует значений $P_i(x)$ и $P_{i-1}(x)$ одновременно.

Видно, что рекуррентную формулу /2.8/ можно использовать для расчета самих значений полинома некоего порядка, если известны рекуррентные коэффициенты и соответствующие значения полиномов двух предшествующих порядков. Это значит, что для построения всей системы полиномов необходимо задать ее первые два члена. Данному требованию легко удовлетворить, если положить

$$P_{-1}(x) \equiv 0 \qquad /2.13/$$

и

$$P_0(x)=1/\beta_0, \qquad /2.14/$$

причем из /2.4/ и /2.5/ следует, что

$$\beta_0=\sqrt{\sum_{k=1}^{m} w_k} \; . \qquad /2.15/$$

Естественно, "полином" P_{-1} введен формально; он нужен только для вычисления $P_1(x)$ по /2.8/, не может удовлетворять /2.4/ и не является членом системы.

В следующем разделе мы воспользуемся /2.8/-/2.15/ для построения алгоритма, вычисляющего рекуррентные коэффициенты α_i и β_i, а также значения полиномов соответствующих степеней в любой точке $x\in[-1,+1]$при заданном множестве M с весами $\{w_i\}$.

3. ПОСТРОЕНИЕ СИСТЕМЫ ОРТОНОРМИРОВАННЫХ ПОЛИНОМОВ

Пусть множество M, определенное согласно /2.1/, задано вместе с соответствующим набором весов $\{w_i\}$. Покажем, что этого достаточно для построения системы полиномов, ортонормированных на M согласно /2.4/ и /2.5/. По существу такое построение является доказательством существования системы. Алгоритм построения описан ниже.

3.1. Упорядочить M и перейти от него к его отображению M′ на $[-1,+1]$ согласно /2.2/. Так как такое преобразование всегда обратимо и, вообще говоря, сделано ради удобства, не являясь при этом существенным для хода рассуждений, мы будем в дальнейшем опускать штрих как знак преобразования.

3.2. Положить $P_{-1}(x)\equiv 0$ согласно /2.13/.

3.3. Вычислить $\beta_0=(\sum_{i=1}^{m} w_j)^{1/2}$ по /2.15/.

3.4. Вычислить $P_0(x)=c_0=1/\beta_0$ по /2.14/. Шаги 3.1-3.4 представляют собой предварительную подготовку к построению системы и выполняются только один раз.

3.5. Положить $i=0$.

3.6. Вычислить α_{i+1} по /2.9/.

3.7. Вычислить $c_{i+1}=1/\beta_{i+1}$ по /2.11/. На этом шаге становится возможным применение рекуррентного соотношения /2.8/, т.е. мы в состоянии вычислить значения $P_{i+1}(x)$ в любой точке $x\in[-1,+1]$.

3.8. Увеличить значение i на единицу.

3.9. Проверить, достигнуто ли предельно допустимое значение i; если оно достигнуто, построение окончено; если нет, то следует вернуться к шагу 3.6.

Вопрос о том, каково число возможных повторений шагов 3.6-3.8,или, иными словами, какова наивысшая степень полиномов в нашей ортонормированной системе, пока остается открытым. Ответом на него является следующая

Теорема 1. Множество полиномов, ортогональных на дискретном точечном множестве M с весами $\{w_i\}_{i=1,2,\dots,m}$, имеет такую же мощность, что и порождающее его точечное множество, и включает в себя полиномы степеней от 0 до $m-1$.

Доказательство. Рассмотрим последовательность полиномов

$$1,\, x,\, x^2,\, x^3,\dots,x^n,\dots \qquad /\text{×}/$$

и покажем, что линейно-независимыми на M с весами $\{w_i\}$ являются только ее первые m членов. Если это действительно так, то равенство $\|\rho\|^2=0$,где $\rho(x)$ - полином степени $r\le m-1$, возможно только в случае $\rho(x)\equiv 0$. В принятой нами метрике

$$\|\rho\|^2 = \sum_{i=1}^{m} \rho^2(x_i)\, w_i \; ,$$

где правая часть может быть нулевой только при $\rho(x)\equiv 0$, так как в противном случае $\rho(x)$ должен был бы иметь более r нулей. Кроме того, начиная с члена x^m и выше, в /×/ проявляется линейная зависимость. Действительно, при $r\ge m$ можно выбрать $\rho(x)$ в виде $q(x)\prod_{i=1}^{m}(x-x_i)$, где $q(x)$ - ненулевой полином степени $r_1=r-m\ge 0$. Тогда

$$\|\rho(x)\|^2=\sum_{i=1}^{m} q^2(x_i)\, w_i \prod_{j=1}^{m}(x_i-x_j)^2\equiv 0,$$

так как в каждом члене присутствует нулевой множитель. Следовательно, в последовательности /×/ линейно-независимы только члены $1, x, x^2, \dots, x^{m-1}$, а из них можно в принципе построить требуемую ортонормированную систему путем ортогонализации.



Блок-схема алгоритма построения системы полиномов, ортонормированных на дискретном множестве M с весами $\{w_i\}_{i=1,2,\dots,m}$. Выходом алгоритма являются значения рекуррентных коэффициентов $\{\alpha_i\}_{i=1,2,\dots,m-1}$ и $\{\beta_i\}_{i=0,1,2,\dots,m-1}$, а также константы $P_{-1}=0$ и $P_0=c_0$.

Кроме того, имеет место и

Теорема 2. Построенная указанным способом система ортонормированных полиномов является полной по отношению к классу функций $f(x)$, для которых точки порождающего множества $\{x_i\}$ - не особые.

Доказательство. Утверждение теоремы эквивалентно тому, что система уравнений

$$f(x_i) = \sum_{j=0}^{m-1} b_j P_j(x_i), \qquad /xx/$$

$$i = 1, 2, \dots, m,$$

имеет единственное решение, определяющее коэффициенты b_j, каковы бы ни были значения компонентов вектора правой части

$$\mathrm{col}\{f(x_1), f(x_2), \dots, f(x_m)\} \in R^m .$$

В самом деле, определитель этой системы

$$D = \begin{vmatrix} P_0(x_1) & P_1(x_1) & \dots & P_{m-1}(x_1) \\ P_0(x_2) & P_1(x_2) & \dots & P_{m-1}(x_2) \\ \dots & \dots & \dots & \dots \\ P_0(x_m) & P_1(x_m) & \dots & P_{m-1}(x_m) \end{vmatrix} \neq 0$$

в силу несовпадения точек порождающего множества M и линейной независимости полиномов ортонормированной системы. Следовательно, система /xx/ имеет единственное решение, чем теорема доказана.

Видно, что несовпадение точек порождающего множества существенно для полноты по отношению к указанному классу функций. В этой связи можно заметить, что хотя предлагаемый алгоритм не накладывает ограничений на распределение точек множества M в интервале $[-1,+1]$, численная его реализация, по-видимому, должна проводиться при точности, адекватной наименьшему расстоянию между двумя соседними точками. На данном этапе проведение количественной оценки степени этой адекватности выходило за рамки нашего исследования.

Теперь мы в состоянии представить алгоритм построения системы полиномов в окончательной блок-схеме /см. рисунок/. Отметим, что построенные таким образом ортонормированные полиномы обладают всеми свойствами, характерными для этого класса функций: ограниченностью, непрерывностью, гладкостью, линейной независимостью, взаимным разделением нулей, свойством наименьшего квадрата и т.д.[/6/].

Несмотря на то, что предложенный алгоритм удобнее всего реализовать на ЭВМ, в относительно простых случаях возможно и аналитическое выполнение всех выкладок. В качестве примера приводятся результаты для случая $m=5$ при значениях

$$\{x_i\} = -1{,}0,\ -0{,}5,\ 0{,}0,\ 0{,}5,\ 1{,}0;$$

и

$$\{w_i\} = 0{,}5,\ 0{,}5,\ 2{,}0,\ 0{,}5,\ 0{,}5.$$

Расчет рекуррентных коэффициентов по описанному алгоритму дает

$$\alpha_1=0{,}0; \quad \alpha_2=0{,}0; \quad \alpha_3=0{,}0; \quad \alpha_4=0{,}0;$$

$$\beta_0=2{,}0; \quad \beta_1=\sqrt{5}/4; \quad \beta_2=\sqrt{43}/4\sqrt{5}; \quad \beta_3=6/\sqrt{215}; \quad \beta_4=\sqrt{10/43}.$$

а попытка расчета β_5 приводит к нулевому результату, как это и должно быть вследствие теоремы 1. Равенство нулю всех α_i легко понять на основе симметрии наборов x_i и w_i; расчет α_5 возможен и тоже приводит к нулевому значению.

Зная β_4 и α_5, мы могли бы рассчитать ненормированный полином $P_5(x)$.

Как и следовало ожидать, все его нули совпадают с точками порождающего множества и нормировка опять-таки оказывается невозможной. Явный вид всех полиномов системы и ненормированного полинома пятой степени приведен ниже:

$$P_0(x) = 1/2,$$

$$P_1(x) = 2x/\sqrt{5},$$

$$P_2(x) = (16x^2 - 5)/2\sqrt{43},$$

$$P_3(x) = (20x^3 - 17x)/3\sqrt{5},$$

$$P_4(x) = (172x^4 - 175x^2 + 9)/3\sqrt{86},$$

$$\overline{P_5}(x) = c_5(x+1)(x+1/2)x(x-1/2)(x-1).$$

Значения полиномов ортонормированной системы в точках порождающего множества M даны в табл. 1.

Таблица 1

Значения полиномов простого примера с $m=5$

x	w	$P_0(x)$	$P_1(x)$	$P_2(x)$	$P_3(x)$	$P_4(x)$
-1,0	0,5	0,5	$-2/\sqrt{5}$	$11/2\sqrt{43}$	$-1/\sqrt{5}$	$2/\sqrt{86}$
-0,5	0,5	0,5	$-1/\sqrt{5}$	$-1/2\sqrt{43}$	$2/\sqrt{5}$	$-8/\sqrt{86}$
0,0	2,0	0,5	0	$-5/2\sqrt{43}$	0	$3/\sqrt{86}$
0,5	0,5	0,5	$1/\sqrt{5}$	$-1/2\sqrt{43}$	$-2/\sqrt{5}$	$-8/\sqrt{86}$
1,0	0,5	0,5	$2/\sqrt{5}$	$11/2\sqrt{43}$	$1/\sqrt{5}$	$2/\sqrt{86}$

Безусловно, табл. 1 можно рассчитать как по явному виду полиномов, так и на основе рекуррентной формулы /2.8/; на наш взгляд, второй путь проще, удобнее и точнее при реализации алгоритма на ЭВМ в целях аппроксимации кривых.

4. ПРОГРАММНАЯ РЕАЛИЗАЦИЯ АЛГОРИТМА

На следующих страницах представлен текст трех программ на ФОРТРАНе- IV, которые реализуют алгоритм построения системы ортонормированных полиномов при заданном множестве M с весами $\{w_i\}$. Последняя /четвертая/ программа является главной и иллюстрирует способ использования этого небольшого пакета из трех SUBROUTINEs в случае примера из предыдущего раздела с $m=5$. Ни одна из программ пакета не резервирует память под рабочие массивы, что должно быть сделано в главной /вызывающей/ программе. Более того, при правильном использовании программы свободны от операций ввода-вывода за исключением печати сообщений об ошибках, могущих возникнуть только при нарушении правил работы с пакетом.

Программа PØRTHN является подготовительной в том смысле, что она рассчитывает значения рекуррентных коэффициентов α_i и β_i. По этой причине ее вызов обязательно должен предшествовать первому рабочему вызову программы ØRTHØN /см. ниже/. Расчет всех необходимых α_i и β_i производится за один вызов PØRTHN; вызывать программу повторно следует только при переходе к новому множеству M, к новым весам $\{w_i\}$ или же при повышении порядка NNMAX нужных ортонормированных полиномов. Список формальных параметров PØRTHN включает:

- M – число точек $\{x_i\}$ порождающего множества;
- NNMAX – максимальную степень полинома, для которого следует вычислить рекуррентные коэффициенты. Как указывалось ранее, NNMAX не должно превышать M-1 ;
- X – упорядоченный одномерный массив значений $\{x_i\}$. Необязательно, чтобы $-1 \le x_i \le +1$, т.к. PØRTHN проводит внутреннее линейное преобразование к этому отрезку;
- W – одномерный массив весов $\{w_i\}$;
- ALPHA – одномерный массив рекуррентных коэффициентов α_i ; α_1, располагается в ALPHA(1) и т.д. до ALPHA(NNMAX) ;
- BETA – одномерный массив рекуррентных коэффициентов β_i , β_0, располагается в BETA(1) , β_1-в BETA(2) и т.д. до BETA(NNMAX+1);

PØLY - одномерный массив значений полиномов при некоем фиксированном значении аргумента X. Этот массив используется здесь как рабочий, при выходе же из PØRTHN в нем находятся значения полиномов в последней точке X(M) множества, причем $P_{-1}=0$ располагается в PØLY(1), P_0 - в PØLY(2), P_1 - в PØLY(3) и т.д. до PØLY(NNMAX+2). Так как P_{-1} и P_0 - постоянные, соответствующие им элементы PØLY в дальнейшем не модифицируются программой ØRTHØN.

Хотя, в принципе, ФОРТРАН не допускает рекурсивных процедур, ØRTHØN использует программу ØRTHØN в циклическом режиме своеобразной неполной рекурсии, когда при каждом выполнении цикла готовятся условия вызова ØRTHØN для последующего его выполнения.

Программа ØRTHØN - основная рабочая программа пакета. Она рассчитывает значения ортонормированных полиномов по рекуррентной формуле /2.8/ при любом значении аргумента, необязательно совпадающем с некоторой из точек порождающего множества. Опять-таки, как это имеет место в PØRTHN, значения аргумента подаются в своих естественных единицах и подвергаются внутреннему линейному преобразованию. Вообще говоря, пользователь пакета может задавать значения аргумента X в привычных для себя единицах и полностью игнорировать внутреннее линейное преобразование. Для правильной работы программ достаточно, чтобы единицы задания аргументов и единицы измерения порождающего множества M были одинаковыми. Конечно, требования упорядоченности множества и положительности весов остаются в силе. ØRTHØN срабатывает при задании следующих формальных параметров:

NNMAX - максимальная степень полинома, чье значение вычисляется в данном вызове. Обозначение этой переменной совпадает с подобной переменной в PØRTHN, но это не значит, что они идентичны. Здесь NNMAX может принимать любое значение, не превосходящее NNMAX из PØRTHN;

XX - значение аргумента ортонормированных полиномов;

ALPHA - то же, что и в PØRTHN; здесь используется, но не модифицируется;

BETA - то же, что и в PØRTHN; здесь используется, но не модифицируется;

PØLY - массив результатов; организован так же, как и в PØRTHN, но соответствует значению аргумента XX.

Программа ERØRTH печатает сообщения об ошибочных ситуациях, обнаруженных при работе PØRTHN и ØRTHØN. Формальных параметров она не имеет и связана с ними через именованный CØMMØN. Ошибочных ситуаций пять, и они соответствуют:

1 - неупорядоченному множеству M;
2 - отрицательному значению NNMAX;
3 - NNMAX, превосходящему M-1;
4 - системному или машинному сбою;
5 - NNMAX в ØRTHØN, превышающему NNMAX в PØRTHN.

Пакет не проверяет положительности набора весов w_i, которая должна обеспечиваться вызывающей программой. Ниже приводятся разности точных значений рекуррентных коэффициентов в примере из предыдущего раздела и соответствующего выхода из четвертой /иллюстрирующей/ программы:

$$\Delta a_1=0;\quad \Delta a_2=5{,}96E-8;\quad \Delta a_3=5{,}96E-8;\quad \Delta a_4=-1{,}79E-7;$$
$$\Delta\beta_0=2{,}0E-7;\ \Delta\beta_1=2{,}84E-7;\ \Delta\beta_2=-3{,}65E-7;\ \Delta\beta_3=-6{,}32E-9;\ \Delta\beta_4=7{,}22E-7.$$

Табл. 2, в которой даны разности $\Delta_i(x)=P_i^{\text{точн.}}(x)-P_i^{\text{расч.}}(x)$, демонстрирует точность работы алгоритма, реализованного в программах. Если учесть, что расчеты были выполнены на малой ЭВМ ЕС-1010, где мантисса чисел с плавающей точкой имеет длину в 24 бит, то точность эту следует признать весьма удовлетворительной.

Таблица 2

x	w	$\Delta_0(x)$	$\Delta_1(x)$	$\Delta_2(x)$	$\Delta_3(x)$	$\Delta_4(x)$
-1,0	0,5	0,0	5,0E-7	-5,0E-7	-6-0E-7	1,6E-6
-0,5	0,5	0,0	3,0E-7	-5,0E-7	-2,0E-7	1,4E-6
0,0	2,0	0,0	0,0	-4,0E-7	-1,0E-7	-1,0E-7
0,5	0,5	0,0	-3,0E-7	-4,0E-7	2,0E-7	2,0E-6
1,0	0,5	0,0	-5,0E-7	-5,0E-7	9,0E-7	1,8E-6

Отметим, что алгоритм построения системы ортонормированных полиномов на точечном множестве M в принципе является точным, откуда следует, что разности таблицы 2 происходят исключительно в результате накопления ошибок округления. В этой связи мы провели дополнительный численный эксперимент по выполнению условий ортогональности /2,4/ на малых ЭВМ ИЗОТ-310 и М-6000, обладающих мантиссой в 23 бит, при m=100 и плавно меняющихся весах. Эксперимент показал, что условие нормировки соблюдается с точностью порядка 2,0E-6 до степеней полиномов 40 и выше, тогда как ортогональность полиномов разных степеней начинает нарушаться несколько раньше, из-за чего сумма /2.4/ получается равной 0,001 при степени 25; ~0,01 при степени 35 и около 0,25 при степени 40. Эти

```
      SUBROUTINE ORTHON(NNMAX,XX,ALPHA,BETA,POLY)
      DIMENSION POLY(1),ALPHA(1),BETA(1)
      COMMON /ORTNRM/X0,X1,MESSG,NPREP,DUMMY
      IF(NPREP-NNMAX)60,10,10
   10 IF(NNMAX)50,40,20
   20 DUMMY=X1*XX+X0
      K=NNMAX+2
      DO 30 I=3,K
      IPOL1=I-1
      IPOL2=IPOL1-1
   30 POLY(I)=((DUMMY-ALPHA(IPOL2))*POLY(IPOL1)-
     1               BETA(IPOL2)*POLY(IPOL2))/BETA(IPOL1)
   40 RETURN
   50 MESSG=2
      CALL ERORTH
      GO TO 40
   60 MESSG=5
      CALL ERORTH
      GO TO 40
      END
      SUBROUTINE ERORTH
      COMMON /ORTNRM/X0,X1,MESSG,NPREP,DUMMY
      GO TO (10,20,30,40,50,40),MESSG
   10 PRINT 110
      GO TO 60
   20 PRINT 120
      GO TO 60
   30 PRINT 130
      GO TO 60
   40 PRINT 140
      GO TO 60
   50 PRINT 150
   60 MESSG=0
      RETURN
  110 FORMAT(' ERROR - SET DISORDERED')
  120 FORMAT(' ERROR - NEGATIVE DEGREE OF POLYNOMIALS REQUESTED')
  130 FORMAT(' ERROR - TOO HIGH DEGREE OF POLYNOMIALS REQUESTED')
  140 FORMAT(' ERROR - SYSTEM AND/OR HARDWARE FALURE')
  150 FORMAT(' ERROR - REQUESTED DEGREE HIGHER THAN THAT PREPARED')
      END
C
C TEST OF ORTHONORMAL POLYNOMIALS OVER 5 POINTS
C
      DIMENSION X(5),W(5),ALPHA(4),BETA(5),POLY(6)
      DATA X/-1.0,-0.5, 0.0, 0.5, 1.0/
      DATA W/ 0.5, 0.5, 2.0, 0.5, 0.5/
      CALL PORTHN(5,4,X,W,ALPHA,BETA,POLY)
      PRINT 104
  104 FORMAT(9X,'I',10X,'ALPHA(I)  ',19X,'J',10X,'BETA(J)')
      DO 5 I=1,5
      I0=I-1
    5 PRINT 105,I,ALPHA(I),I0,BETA(I)
  105 FORMAT(I10,F20.10,I20,F20.10)
      PRINT 101
      DO 10 I=1,5
      CALL ORTHON(4,X(I),ALPHA,BETA,POLY)
   10 PRINT 100,X(I),W(I),(POLY(II+1),II=1,5)
  100 FORMAT(2F10.1,5F12.7)
  101 FORMAT(32H        X          W          P0(X),
     1         48H       P1(X)          P2(X)          P3(X)          P4(X)   /
     2         32H   =============================,
     3         48H================================================ )
      STOP
      END
```



```
      SUBROUTINE PORTHN(M,NNMAX,X,W,ALPHA,BETA,POLY)
      DIMENSION X(1),W(1),ALPHA(1),BETA(1),POLY(1)
      COMMON /ORTNRM/X0,X1,MESSG,NPREP,TR
C
C CHECK WHETHER SET IS ORDERED
C
      M1=M-1
      DO 5 I=1,M1
      IF(X(I+1)-X(I))710,710,5
    5 CONTINUE
C
C LINEAR TRANSFORMATION TO [-1,1]
C
      X0=X(M)-X(1)
      X1=2.0/X0
      X0=-(X(M)+X(1))/X0
C
C SET POLYNOMIAL VALUE WHEN DEGREE IS -1
C
      POLY(1)=0.0
      IF(NNMAX)720,15,15
   15 NPREP=NNMAX
      IF(M-1-NNMAX)730,20,20
C
C ZERO-DEGREE NORMALIZING FACTOR & VALUE
C
   20 BETA(1)=0.0
      DO 30 I=1,M
   30 BETA(1)=BETA(1)+W(I)
      BETA(1)=SQRT(BETA(1))
      POLY(2)=1.0/BETA(1)
      IF(NNMAX)740,900,40
C
C HIGHER DEGREE RECURRENCE FACTORS
C
   40 DO 100 I=1,NNMAX
      IB=I+1
      IPOL=I-1
      ALPHA(I)=0.0
      BETA(IB)=0.0
      DO 50 K=1,M
      CALL ORTHON(IPOL,X(K),ALPHA,BETA,POLY)
      TR=X1*X(K)+X0
      BETA(IB)=BETA(IB)+(TR**2)*(POLY(IB)**2)*W(K)
   50 ALPHA(I)=ALPHA(I)+TR*W(K)*(POLY(IB)**2)
      BETA(IB)=BETA(IB)-ALPHA(I)**2
      IF(I-1)60,60,55
   55 BETA(IB)=BETA(IB)-BETA(I)**2
   60 BETA(IB)=SQRT(BETA(IB))
  100 CONTINUE
      GO TO 900
  710 MESSG=1
      GO TO 800
  720 MESSG=2
      GO TO 800
  730 MESSG=3
      GO TO 800
  740 MESSG=4
  800 CALL ERORTH
  900 RETURN
      END
```

данные соответствуют максимальному значению суммы /2.4/ по всем полиномам порядка ниже указанного. Можно ожидать, что выполнение расчетов с двойной точностью расширит диапазон степеней, для которых точность выполнения условий /2.4/ и /2.5/ является приемлемой. Впрочем, во многих практических случаях это обстоятельство представляет собой только академический интерес, поскольку применение полиномов таких высоких степеней оказывается ненужным.

5. ЗАКЛЮЧЕНИЕ

Построение системы полиномов, ортонормированных на заданном дискретном множестве, не является самоцелью. Мы разделяем мнение авторов [6,8,9] о том, что численная аппроксимация измеренных зависимостей рядом по ортогональным полиномам имеет значительные преимущества по сравнению с обычными полиномиальными рядами. Тем не менее нам не известны ни сведения о программной реализации метода на ЭВМ, ни о его использовании в практике обработки данных.

В следующей работе мы собираемся рассмотреть в деталях способ использования ортонормированных полиномов на примере нескольких физических задач. Хотелось бы верить, что это приведет к большей популярности данного метода среди физиков.

ЛИТЕРАТУРА

1. Джексон Д. Ряды Фурье и ортогональные полиномы. ИЛ, М., 1948.
2. Толстов Г.П. Ряды Фурье. Гостехиздат, М.-Л., 1951.
3. Сеге С. Ортогональные многочлены. Физматгиз, М., 1962.
4. Зигмунд А. Тригонометрические ряды, тт. 1 и 2. "Мир", М., 1965.
5. Жижиашвили Л.В. Некоторые вопросы теории рядов Фурье и сопряженных к ним тригонометрических рядов. Изд-во ТГУ, 1965.
6. Beckmann P. Orthogonal Polynomials for Engineers and Physicists. The Golem Press, Boulder(Colorado,USA), 1973.
7. Суетин П.К. Классические ортогональные многочлены. "Наука", М., 1979.
8. Forsythe G.E. J.Soc.Industr.Appl. Math., 1957, 5, pp.74-88.
9. Худсон Д. Статистика для физиков. "Мир", М., 1967.

Рукопись поступила в издательский отдел
12 октября 1979 года.